MAXIMUM LIKELIHOOD ESTIMATION IN MPLUS

EMPLOYEE DATA

- Data set containing scores from 480 employees on eight work-related variables
- Variables:
- Age, gender, job tenure, IQ, psychological well-being, job satisfaction, job performance, and turnover intentions
- 33% of the cases have missing well-being scores, and 33% have missing satisfaction scores

ANALYSIS EXAMPLE

- Multiple regression model that predicts job performance from psychological well-being and job satisfaction

$$jobperf = B_0 + B_1(wbeing) + B_2(jobsat) + \varepsilon$$



MPLUS COMMANDS

- TITLE
- DATA
- VARIABLE
- ANALYSIS
- MODEL
- MODEL TEST
- OUTPUT

A FEW MPLUS RULES

- Capitalization never matters
- Variable names must be 8 characters or less
- Command lines must be less than 80 characters in length, wrap commands to the next line as needed
- ! to comment out a line that you want the program to ignore
- : at the end of a command
- ; at the end of a subcommand

TITLE COMMAND

- The TITLE command (optional) prints a title on output file

```
TITLE:
! The title command is optional;
mplus multiple regression program;
```

DATA COMMAND

- The DATA command points Mplus to the location of the text data on the local drive
- Free format text files end in .dat or .txt and should include a placeholder for missing values

```
DATA:
! Location of the data file;
file = 'c:\Data\employee.dat';
```

ALTERNATE DATA COMMAND

- Omit the file path when the data file and the Mplus syntax file are located in the same folder

```
DATA:
! Location of the data file;
file = employee.dat;
```

VARIABLE COMMAND

- The VARIABLE command (a) gives the order of the variables in the data file, (b) selects variables for analysis, and (c) gives the missing value code

```
VARIABLE:
! Information about the contents of the data file;
names = id age tenure female wbeing jobsat jobperf turnover iq;
usevariables = wbeing jobsat jobperf;
missing = all (-99);
```

ANALYSIS COMMAND

- ANALYSIS specifies the estimator and other estimation details

```
ANALYSIS:
! Specify the estimator (ML is usually the default);
estimator = ml;
```

MODEL COMMAND

- The MODEL command specifies the analysis
- Mplus automatically estimates many parameters (e.g., variances, residual variances, means)
- Missing data models can require additional parameters

```
MODEL:
! Regression model – "on" means "regressed on";
jobperf on wbeing jobsat;
```

INCOMPLETE PREDICTOR VARIABLES

- The missing data log likelihood always allows for incomplete data on *Y* (i.e., outcome) variables
- Software packages that implement ML estimation generally exclude cases with incomplete data on manifest predictor variables
- SEM packages provide a mechanism for dealing with incomplete predictors

SEM SPECIFICATION

- A latent variable replaces each incomplete predictor, and the predictor becomes an outcome of its replacement latent variable (i.e., a *Y*)
- Step 1: Define each incomplete predictor variable as the sole indicator of a latent variable
- Step 2: Fix the factor loading for each latent / predictor to 1
- Step 3: Fix the residual variance of each indicator to 0 (or something close to zero, e.g., .0001)
- Step 4: Correlate each latent variable with all other predictors (manifest or latent)

PATH DIAGRAM

- The latent variables are exact duplicates of the manifest variables (i.e., have same mean, variance, correlation)
- The interpretation of B_1 and B_2 does not change!



MODIFYING THE MODEL COMMAND

- Specify the variance of each incomplete predictor as well as its covariance with all other predictors (manifest or latent)
- This is a shorthand way of specifying the latent variable model

```
MODEL:
jobperf on wbeing jobsat;  ! Regression;
wbeing jobsat;             ! Variances of IVs;
wbeing with jobsat;        ! Covariance between IVs;
```

WALD TEST

- In ML analyses, Wald chi-square statistics are routinely used to test a set of parameters for significance
- The single-parameter version of the test is as follows

$$\omega = \frac{\left(\hat{\theta} - \theta_0\right)^2}{SE^2}$$

- The Wald test is the ML analog of an *F* statistic in OLS regression or ANOVA

THE WALD TEST IN MPLUS

- Wald test parameters must have labels
- The label in parentheses is arbitrary

```
MODEL:
! (b1) and (b2) labels that are used to specify custom hypotheses;
jobperf on wbeing (b1);
jobperf on jobsat (b2);
```

MODEL TEST COMMAND

- The MODEL TEST generates a Wald test for many custom hypotheses

```
MODEL TEST:
! Two df omnibus test where both coefficients = 0;
! b1 and b2 are user-supplied labels from MODEL;
b1 = 0;
b2 = 0;
```

OUTPUT COMMAND

- The OUTPUT command specifies optional information that appears in the Mplus output file

```
OUTPUT:
! standardized gives beta weights and R-square;
! sampstat gives ML descriptives;
! patterns prints missing data patterns;
standardized sampstat patterns;
```

MPLUS REGRESSION PROGRAM

```
DATA:
file = employee.dat;
VARIABLE:
names = id age tenure female wbeing jobsat jobperf turnover iq;
usevariables = wbeing jobsat jobperf;
missing = all (-99);
ANALYSIS:
estimator = ml;
MODEL:
jobperf on wbeing (b1);
jobperf on jobsat (b2);
wbeing jobsat;
wbeing with jobsat;
MODEL TEST:
b1 = 0;
b2 = 0;
OUTPUT:
standardized sampstat patterns;
```

MISSING DATA PATTERNS (PATTERNS OPTION)

```
SUMMARY OF MISSING DATA PATTERNS

    MISSING DATA PATTERNS (x = not missing)

          1  2  3
JOBPERF   x  x  x
WBEING    x  x
JOBSAT    x     x

     MISSING DATA PATTERN FREQUENCIES

Pattern   Frequency     Pattern   Frequency     Pattern   Frequency
      1         160           2         160           3         160
```

COVARIANCE COVERAGE

- The covariance coverage matrix gives the proportion of complete cases on each variable or variable pair

```
PROPORTION OF DATA PRESENT

           Covariance Coverage
              JOBPERF       WBEING        JOBSAT
              _______       _______       _______
JOBPERF         1.000
WBEING          0.667         0.667
JOBSAT          0.667         0.333         0.667
```

DESCRIPTIVES (SAMPSTAT OPTION)

ESTIMATED SAMPLE STATISTICS

Means

	JOBPERF	WBEING	JOBSAT
	6.021	6.286	5.959

Covariances

	JOBPERF	WBEING	JOBSAT
JOBPERF	1.570		
WBEING	0.673	1.387	
JOBSAT	0.259	0.466	1.390

Correlations

	JOBPERF	WBEING	JOBSAT
JOBPERF	1.000		
WBEING	0.456	1.000	
JOBSAT	0.175	0.335	1.000

WALD TEST (MODEL TEST COMMAND)

- The Wald statistic (a chi-square with 2 degrees of freedom) is akin to the omnibus *F* test in OLS regression

Wald Test of Parameter Constraints

Value	95.882
Degrees of Freedom	2
P-Value	0.0000

- The significant chi-square, $\chi^2(2) = 95.882$, indicates that the set of predictors explain significant variation in the dependent variable

UNSTANDARDIZED ESTIMATES

MODEL RESULTS

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
JOBPERF ON				
WBEING	0.476	0.055	8.665	0.000
JOBSAT	0.027	0.060	0.444	0.657
WBEING WITH				
JOBSAT	0.467	0.098	4.780	0.000
Means				
WBEING	6.286	0.063	99.692	0.000

INTERPRETATIONS

- Interpret and report ML estimates in the same way as a complete-data analysis
- Controlling for job satisfaction, a one-point increase in psychological well-being results in a .476 increase in job performance, on average
- Controlling for psychological well-being, a one-point increase job satisfaction in results in a .027 increase in job performance, on average

UNSTANDARDIZED ESTIMATES, CONTINUED

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Intercepts				
JOBPERF	2.869	0.382	7.517	0.000
Variances				
WBEING	1.387	0.108	12.852	0.000
JOBSAT	1.390	0.109	12.711	0.000
Residual Variances				
JOBPERF	1.243	0.087	14.356	0.000

STANDARDIZED ESTIMATES (STANDARDIZED OPTION)

STANDARDIZED MODEL RESULTS

STDYX Standardization

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
JOBPERF ON				
WBEING	0.447	0.049	9.181	0.000
JOBSAT	0.025	0.056	0.444	0.657

STANDARDIZED ESTIMATES, CONTINUED

R-SQUARE

Observed Variable	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
JOBPERF	0.208	0.039	5.350	0.000

INTERPRETATIONS

- The STDYX standardization gives beta weights
  - Controlling for job satisfaction, a one standard deviation increase in psychological well-being results in a .447 standard deviation increase in job performance, on average
  - Controlling for psychological well-being, a one standard deviation increase job satisfaction in results in a .025 standard deviation increase in job performance, on average
- Together, the two predictors explain 20.8% of the variance in job performance ratings